A Computational Verification of Annihilator-Based Radical Dependency via OSCAR

Hamdullah Özkaya
Department of Mathematics, Bursa Technical University, Bursa, Turkey
hamdullah.ozkaya@btu.edu.tr

July 21, 2026

Abstract

This note provides an independent computational verification, using the OSCAR computer algebra system, of six theorems from a recent paper on annihilator-based dependency relations for submodules of a Noetherian module. We reproduce the paper's elementary examples over $\mathbb{Z}/n\mathbb{Z}$, test the characterization of radical dependence on explicit polynomial-ring modules with matching and with distinct associated primes, verify total annihilator-dependence and its necessity under a dimension-distinctness hypothesis, construct via localization a module whose Radical Distinction Set is empty by design, confirm the universal radical sum equivalence across several submodule pairs, and check the multiplication-module identity exhaustively on the full submodule lattice of $\mathbb{Z}/25\mathbb{Z}$. In every case the computed annihilators and radicals agree exactly with the theoretical predictions.

Keywords: annihilator ideals, radical submodules, associated primes, multiplication modules, computer algebra, OSCAR, Julia.

2020 Mathematics Subject Classification: 13C13, 13C05, 13P99.

1 Introduction

In [1], the authors introduce a notion of radical dependence for submodules N_1, N_2 of a module M over a commutative Noetherian ring R: N_1 and N_2 are *radically dependent* if

$$\sqrt{\mathrm{Ann}(N_1 + N_2)} = \sqrt{\mathrm{Ann}(N_1)} + \sqrt{\mathrm{Ann}(N_2)},$$

and it is shown (Theorem 1 of [1]) that this holds if and only if

$$\sqrt{\mathrm{Ann}(N_1)} = \sqrt{\mathrm{Ann}(N_2)}.$$

The purpose of this short note is purely computational: we use the OSCAR system [2], written in Julia and combining Singular, GAP, Polymake and Antic, to construct explicit modules and submodules, compute their annihilators and radicals via Gröbner basis and primary decomposition algorithms, and check the identity above symbolically rather than by hand. All computations below were carried out with OSCAR (Julia 1.12, `Oscar.jl` as installed from the General registry in July 2026) and can be reproduced verbatim by pasting the listed code into a Julia session after `using Oscar`. The complete, individually runnable source code for every computation in this note, together with a `Project.toml` pinning the OSCAR dependency and instructions for generating a

`Manifest.toml` with the exact resolved package versions on the reader's own machine, is available in the accompanying repository.[1]

2 Elementary examples over $\mathbb{Z}$

Examples 1–3 of [1] concern $M = \mathbb{Z}/n\mathbb{Z}$ as a $\mathbb{Z}$-module for $n = 8, 5, 25$. Since $\mathbb{Z}$ is a principal ideal domain, annihilators of cyclic submodules are computed directly from gcd, which we record here for completeness.

```
n = 8
for m in 1:(n-1)
    a = n / gcd(m, n)
    println("Ann(", m, "-bar) = (", Int(a), ")Z")
end
```

This reproduces $\mathrm{Ann}(\bar{1}) = \mathrm{Ann}(\bar{3}) = \mathrm{Ann}(\bar{5}) = \mathrm{Ann}(\bar{7}) = 8\mathbb{Z}$, $\mathrm{Ann}(\bar{2}) = \mathrm{Ann}(\bar{6}) = 4\mathbb{Z}$, and $\mathrm{Ann}(\bar{4}) = 2\mathbb{Z}$, exactly as in Example 1 of [1]. The same computation with $n = 5$ and $n = 25$ reproduces Examples 2 and 3 respectively.

3 Verification of Theorem 1 over a polynomial ring

For reference, we restate the result verbatim from [1].

Theorem 1 (Characterization of radical dependence; Theorem 1 of [1])**.** *Let $N_1, N_2 \leq M$. Then N_1 and N_2 are radically dependent if and only if $\sqrt{\mathrm{Ann}(N_1)} = \sqrt{\mathrm{Ann}(N_2)}$.*

To test the characterization of radical dependence in a setting with a non-principal-ideal-domain base ring, we work over $R = \mathbb{Q}[x, y]$ and realize each submodule as an explicit `SubquoModule` of a rank-one free module, following the construction used in the proof of Theorem 3 of [1]: given a prime ideal $\mathfrak{p} \subset R$, the cyclic module $R/\mathfrak{p}$ is represented as $F/\mathfrak{p}F$ for $F \cong R$, and $M = R/\mathfrak{p}_1 \oplus R/\mathfrak{p}_2$ is formed as a direct sum, with $N_1, N_2 \leq M$ the two canonical summands. For this construction $\mathrm{Ann}(N_i) = \mathfrak{p}_i$ and $\mathrm{Ann}(N_1 + N_2) = \mathrm{Ann}(N_1) \cap \mathrm{Ann}(N_2)$.

3.1 Case $\mathfrak{p}_1 \neq \mathfrak{p}_2$

Take $\mathfrak{p}_1 = (x)$ and $\mathfrak{p}_2 = (x - 1, y)$, two distinct prime ideals of $R = \mathbb{Q}[x, y]$.

```
R, (x, y) = polynomial_ring(QQ, [:x, :y])

p1 = ideal(R, [x])
p2 = ideal(R, [x - 1, y])

F1 = free_module(R, 1)
N1_alone = SubquoModule(F1, R[one(R);], R[x;])

F2 = free_module(R, 1)
N2_alone = SubquoModule(F2, R[one(R);], R[x-1; y])

M, incs = direct_sum(N1_alone, N2_alone)
```

[1] https://github.com/REPLACE-WITH-USERNAME/annihilator-dependency-oscar-verification (placeholder – to be created and filled in before submission).

```
N1, _ = image(incs[1])
N2, _ = image(incs[2])

J1    = annihilator(N1)
J2    = annihilator(N2)
Jsum  = annihilator(N1 + N2)

rad_J1   = radical(J1)
rad_J2   = radical(J2)
rad_Jsum = radical(Jsum)

println(rad_J1 == p1)                              # true
println(rad_J2 == p2)                              # true
println(rad_Jsum == intersect(rad_J1, rad_J2))      # true
println(rad_Jsum == rad_J1 + rad_J2)                # false
```

The computation returns `rad_J1 == p1` and `rad_J2 == p2`, confirming that the constructed submodules indeed have the intended annihilators. It further confirms $\sqrt{\mathrm{Ann}(N_1+N_2)} = \sqrt{\mathrm{Ann}(N_1)} \cap \sqrt{\mathrm{Ann}(N_2)}$ (always true, by the identity $\mathrm{Ann}(N_1+N_2) = \mathrm{Ann}(N_1) \cap \mathrm{Ann}(N_2)$ used in the proof of Theorem 1), while $\sqrt{\mathrm{Ann}(N_1+N_2)} \neq \sqrt{\mathrm{Ann}(N_1)} + \sqrt{\mathrm{Ann}(N_2)}$: since $\mathfrak{p}_1 \neq \mathfrak{p}_2$ are distinct maximal-type primes with $\mathfrak{p}_1 + \mathfrak{p}_2 = R$ (their associated varieties, the line $x = 0$ and the point $(1, 0)$, are disjoint), the sum $\sqrt{J_1} + \sqrt{J_2}$ is the unit ideal, while $\sqrt{J_{\mathrm{sum}}}$ is a proper ideal. This is precisely the failure of radical dependence predicted by Theorem 1 when $\sqrt{\mathrm{Ann}(N_1)} \neq \sqrt{\mathrm{Ann}(N_2)}$.

3.2 Case $\mathfrak{p}_1 = \mathfrak{p}_2$

Repeating the construction with $\mathfrak{p}_1 = \mathfrak{p}_2 = (x)$:

```
p1 = ideal(R, [x])
p2 = ideal(R, [x])

F1 = free_module(R, 1)
N1_alone = SubquoModule(F1, R[one(R);], R[x;])
F2 = free_module(R, 1)
N2_alone = SubquoModule(F2, R[one(R);], R[x;])

M, incs = direct_sum(N1_alone, N2_alone)
N1, _ = image(incs[1])
N2, _ = image(incs[2])

J1   = annihilator(N1);   J2   = annihilator(N2)
Jsum = annihilator(N1 + N2)

rad_J1   = radical(J1)
rad_J2   = radical(J2)
rad_Jsum = radical(Jsum)

println(rad_Jsum == rad_J1 + rad_J2)   # true
```

Here the output is `true`: with $\mathfrak{p}_1 = \mathfrak{p}_2$, the radical dependence identity of Theorem 1 holds exactly, as predicted.

4 Verification of Theorem 2 (total annihilator-dependence)

Theorem 2 (Theorem 2 of [1]). *Let R be a commutative Noetherian ring with identity, and let M be a finitely generated R-module. If $\mathrm{Ass}(M) = \{\mathfrak{p}\}$, then M is totally annihilator-dependent, i.e. for all nonzero submodules $N_1, N_2 \le M$ with $\mathrm{Ann}(N_i) \neq 0$,*

$$\dim R/\operatorname{Ann}(N_1 + N_2) = \min\{\dim R/\operatorname{Ann}(N_1),\ \dim R/\operatorname{Ann}(N_2)\}.$$

To test this on a module whose associated submodules genuinely have *different* annihilators (so that the identity is not trivially true by equality of ideals), we take $R = \mathbb{Q}[x, y]$ and $M = R/(x^2)$, whose defining ideal (x^2) is already primary with associated prime (x); hence $\mathrm{Ass}(M) = \{(x)\}$ is a singleton, as required by the hypothesis. Inside M we consider the cyclic submodules $N_1 = R\bar{x}$ and $N_2 = R\bar{y}$ generated by the images of x and y respectively.

In OSCAR, the associated primes of an ideal are extracted from `primary_decomposition`, which returns a vector of (primary, prime) pairs; there is no ideal-level `associated_primes` function in the version used here.

```
R, (x, y) = polynomial_ring(QQ, [:x, :y])

I = ideal(R, [x^2])

pd = primary_decomposition(I)
println("Ass(R/I) = ", [P for (Q, P) in pd])   # {(x)}, singleton

F = free_module(R, 1)
Mmod = SubquoModule(F, R[one(R);], R[x^2;])   # M = R/(x^2)
g = gens(Mmod)[1]

N1, _ = sub(Mmod, [x*g])   # R * xbar
N2, _ = sub(Mmod, [y*g])   # R * ybar

J1   = annihilator(N1)
J2   = annihilator(N2)
Jsum = annihilator(N1 + N2)

println("J1 = ", J1, "   radical = ", radical(J1))
println("J2 = ", J2, "   radical = ", radical(J2))
println("Jsum = ", Jsum, " radical = ", radical(Jsum))

A1,   _ = quo(R, J1)
A2,   _ = quo(R, J2)
Asum, _ = quo(R, Jsum)

d1, d2, dsum = dim(A1), dim(A2), dim(Asum)
println("dim R/J1 = ", d1, ", dim R/J2 = ", d2, ", dim R/Jsum = ", dsum)
println("Theorem 2 holds? ", dsum == min(d1, d2))
```

The computation confirms `Ass(R/I) = {(x)}`, verifying the hypothesis rather than merely assuming it. It further returns $J_1 = \mathrm{Ann}(N_1) = (x)$ and $J_2 = \mathrm{Ann}(N_2) = (x^2)$: these two submodules have *genuinely different* annihilators, so the equality $\dim R/J_{\mathrm{sum}} = \min\{\dim R/J_1, \dim R/J_2\}$ is not a tautology. Since $\sqrt{J_1} = \sqrt{J_2} = (x)$, both quotient rings have Krull dimension 1, and OSCAR confirms $\dim R/J_{\mathrm{sum}} = 1 = \min\{1, 1\}$ as well, so the predicted equality holds exactly.

5 Verification of Theorem 3 (necessity of a singleton $\mathrm{Ass}(M)$)

Theorem 3 (Theorem 3 of [1])**.** *Let R be a commutative Noetherian ring with identity, and let M be a finitely generated R-module. Suppose that for $\mathfrak{p}_1, \mathfrak{p}_2 \in \mathrm{Ass}(M)$, whenever $\mathfrak{p}_1 \neq \mathfrak{p}_2$, one has $\dim R/\mathfrak{p}_1 \neq \dim R/\mathfrak{p}_2$. If M is totally annihilator-dependent, then $\mathrm{Ass}(M)$ is a singleton.*

We verify the contrapositive: we exhibit a module M with $\mathrm{Ass}(M) = \{\mathfrak{p}_1, \mathfrak{p}_2\}$, $\mathfrak{p}_1 \neq \mathfrak{p}_2$, satisfying the dimension-distinctness hypothesis, and show computationally that M is *not* totally annihilator-dependent, by exhibiting a witnessing pair N_1, N_2 for which the defining identity of Definition 2 of [1] fails.

We reuse the construction of Section 3.1: $\mathfrak{p}_1 = (x)$ (a line, $\dim R/\mathfrak{p}_1 = 1$) and $\mathfrak{p}_2 = (x-1, y)$ (a point, $\dim R/\mathfrak{p}_2 = 0$), so the hypothesis $\dim R/\mathfrak{p}_1 \neq \dim R/\mathfrak{p}_2$ holds, and $M = R/\mathfrak{p}_1 \oplus R/\mathfrak{p}_2$ has $\mathrm{Ass}(M) = \{\mathfrak{p}_1, \mathfrak{p}_2\}$ via the two summands N_1, N_2.

```
R, (x, y) = polynomial_ring(QQ, [:x, :y])

p1 = ideal(R, [x])
p2 = ideal(R, [x - 1, y])

F1 = free_module(R, 1)
N1_alone = SubquoModule(F1, R[one(R);], R[x;])
F2 = free_module(R, 1)
N2_alone = SubquoModule(F2, R[one(R);], R[x-1; y])

M, incs = direct_sum(N1_alone, N2_alone)
N1, _ = image(incs[1])
N2, _ = image(incs[2])

J1   = annihilator(N1)
J2   = annihilator(N2)
Jsum = annihilator(N1 + N2)

A1,   _ = quo(R, J1)
A2,   _ = quo(R, J2)
Asum, _ = quo(R, Jsum)

d1, d2, dsum = dim(A1), dim(A2), dim(Asum)

println("dim R/p1 = ", d1)
println("dim R/p2 = ", d2)
println("dim R/(p1 cap p2) = ", dsum)
println("Hypothesis of Theorem 3 (d1 != d2)? ", d1 != d2)
println("dsum == max(d1,d2)? ", dsum == max(d1, d2))
println("dsum == min(d1,d2)? ", dsum == min(d1, d2))
```

OSCAR returns $d_1 = 1$, $d_2 = 0$, and $d_{\mathrm{sum}} = 1$. Thus $d_{\mathrm{sum}} = \max\{d_1, d_2\}$, confirming the standard fact that $\dim R/(\mathfrak{p}_1 \cap \mathfrak{p}_2) = \max\{\dim R/\mathfrak{p}_1, \dim R/\mathfrak{p}_2\}$ used in the proof of Theorem 3 in [1], and *not* $\min\{d_1, d_2\}$. This is exactly the failure of the total annihilator-dependence identity: since $d_1 \neq d_2$ (the theorem's hypothesis is met) and $\mathrm{Ass}(M)$ has two elements, M fails to be totally annihilator-dependent, precisely as the contrapositive of Theorem 3 requires. Equivalently, this exhibits an explicit obstruction: a singleton $\mathrm{Ass}(M)$ is not merely sufficient (Theorem 2) but, under the dimension-distinctness hypothesis, necessary for total annihilator-dependence.

6 Verification of Theorem 4 (the Radical Distinction Set)

Definition 1 (Radical Distinction Set; Definition 3 of [1])**.** *Let M be an R-module. The Radical Distinction Set of M is*

$$Z_g(M) = \{\, m \in M \setminus \{0\} \mid \operatorname{Ann}(Rm) \neq \operatorname{Ann}(\operatorname{Rad}(M)) \,\}.$$

Theorem 4 (Theorem 4 of [1])**.** *Let R be a Noetherian commutative ring and $M \neq 0$ a finitely generated R-module. If $Z_g(M) = \emptyset$, then $\operatorname{Ass}(M) = \{\operatorname{Ann}(\operatorname{Rad}(M))\}$, and $\operatorname{Ann}(\operatorname{Rad}(M))$ is a prime ideal.*

OSCAR has no built-in `Rad` or `Zg` function, so both must be assembled from primitives. A first obstacle is that neither of the elementary examples of [1] (Examples 1 and 2, i.e. $\mathbb{Z}/8\mathbb{Z}$ and $\mathbb{Z}/5\mathbb{Z}$) actually satisfies the hypothesis $Z_g(M) = \emptyset$: in both cases $Z_g(M) \neq \emptyset$, so Theorem 4 is never triggered by the paper's own illustrative examples. To exercise it meaningfully we construct a fresh module for which $Z_g(M) = \emptyset$ can be established outright, rather than merely assumed.

Take $R_0 = \mathbb{Q}[x]$ and localize at the maximal ideal of the origin, $U = R_0 \setminus (x)$, to obtain the local domain $R = R_{(x)}$ with unique maximal ideal $\mathfrak{m} = (x)R$. Let $M = R$, viewed as a module over itself. Since R is local, $\operatorname{Rad}(M) = \mathfrak{m}$ (the unique maximal submodule). Since R is furthermore a *domain*, multiplication by any nonzero ring element is injective, so $\operatorname{Ann}(Rm) = (0)$ for every nonzero $m \in M$, and likewise $\operatorname{Ann}(\mathfrak{m}) = (0)$ (only 0 can kill a nonzero ideal of a domain). Hence *every* nonzero m satisfies $\operatorname{Ann}(Rm) = \operatorname{Ann}(\operatorname{Rad}(M)) = (0)$, so $Z_g(M) = \emptyset$ by a structural argument – not by an exhaustive search, which would be impossible here since R is infinite.

```
R0, (x,) = polynomial_ring(QQ, [:x])
U = complement_of_point_ideal(R0, [0])
RL, iota = localization(R0, U)

xL = iota(x)

F = free_module(RL, 1)
Nm, _ = sub(F, [xL*F[1]])      # the maximal ideal m as a submodule of M = RL
Ann_m = annihilator(Nm)
println("Ann(Rad(M)) = ", Ann_m)

zero_ideal = ideal(RL, [zero(RL)])
pd = primary_decomposition(zero_ideal)
println("Ass(M) = ", [P for (Q, P) in pd])

println("Ann(Rad(M)) == (0)? ", Ann_m == zero_ideal)
println("Theorem 4 holds? Ass(M) = {Ann(Rad(M))}? ",
        length(pd) == 1 && pd[1][2] == Ann_m)
```

OSCAR confirms $\operatorname{Ann}(\operatorname{Rad}(M)) = (0)$, and `primary_decomposition` of the zero ideal returns the single pair $((0), (0))$, so $\operatorname{Ass}(M) = \{(0)\}$. Both final checks evaluate to `true`: the predicted equality $\operatorname{Ass}(M) = \{\operatorname{Ann}(\operatorname{Rad}(M))\}$ holds exactly, and the ideal (0) is indeed prime, as required by Theorem 4.

Remark 1. *This example is complementary to the module $M = R/(x^2)$ of Section 4: that module also has a singleton $\operatorname{Ass}(M) = \{(x)\}$, yet $Z_g(M) \neq \emptyset$ there (the submodules $R\bar{x}$ and $R\bar{y}$ have different annihilators $(x) \neq (x^2)$, neither equal to $\operatorname{Ann}(\operatorname{Rad}(M))$ in general). This shows that the hypothesis of Theorem 4 is sufficient, but not necessary, for $\operatorname{Ass}(M)$ to be a singleton – consistent with Theorem 2 providing a strictly weaker sufficient condition (singleton $\operatorname{Ass}(M)$ alone) for the related but distinct conclusion of total annihilator-dependence.*

7 Verification of Theorem 5 (radical sum equivalence)

Theorem 5 (Theorem 5 of [1]). *Let R be a Noetherian commutative ring and M a finitely generated R-module. The following are equivalent:*

(i) for all nonzero $N_1, N_2 \leq M$: $\sqrt{\mathrm{Ann}(N_1 + N_2)} = \sqrt{\mathrm{Ann}(N_1)} + \sqrt{\mathrm{Ann}(N_2)}$;

(ii) $\mathrm{Ass}(M) = \{\mathfrak{p}\}$ for some prime ideal $\mathfrak{p}$.

Theorem 5 strengthens Theorem 1 from a statement about a *single* pair N_1, N_2 to a statement about *all* pairs simultaneously. The direction (i)⇒(ii) is exactly the contrapositive already exhibited in Section 3.1: there, with $\mathrm{Ass}(M)$ containing the two distinct primes $\mathfrak{p}_1 = (x)$ and $\mathfrak{p}_2 = (x-1, y)$, a single witnessing pair N_1, N_2 was already enough to falsify the radical sum identity, confirming ¬(i) whenever ¬(ii) holds – no further computation is needed for that direction.

For (ii)⇒(i), we reuse the module $M = R/(x^2)$ of Section 4, whose associated primes were confirmed to be the singleton $\{(x)\}$ via `primary_decomposition`. Since a single instance cannot establish a claim about *all* pairs N_1, N_2, we test the identity on four structurally distinct pairs of cyclic submodules, including pairs whose annihilators $\mathrm{Ann}(N_1)$ and $\mathrm{Ann}(N_2)$ are themselves unequal.

```
R, (x, y) = polynomial_ring(QQ, [:x, :y])
I = ideal(R, [x^2])

pd = primary_decomposition(I)
println("Ass(R/I) = ", [P for (Q, P) in pd])   # {(x)}

F = free_module(R, 1)
Mmod = SubquoModule(F, R[one(R);], R[x^2;])
g = gens(Mmod)[1]

pairs = [
    (x*g, y*g),
    (y*g, (x+y)*g),
    ((2*x+3*y)*g, (x-y)*g),
    (x*g, (x+y)*g),
]

for (i, (a, b)) in enumerate(pairs)
    Na, _ = sub(Mmod, [a])
    Nb, _ = sub(Mmod, [b])
    Ja   = annihilator(Na)
    Jb   = annihilator(Nb)
    Jsum = annihilator(Na + Nb)
    lhs = radical(Jsum)
    rhs = radical(Ja) + radical(Jb)
    println("Pair $i: J_a=", Ja, "  J_b=", Jb, "  lhs==rhs? ", lhs == rhs)
end
```

All four pairs return `lhs==rhs? true`, including Pairs 1 and 4, in which $\mathrm{Ann}(N_1) = (x)$ and $\mathrm{Ann}(N_2) = (x^2)$ are themselves *unequal* ideals: the radical sum identity still holds because both annihilators share the same radical $(x) = \sqrt{\mathrm{Ann}(M)}$, consistent with Theorem 1. While a finite sample of pairs cannot constitute a proof of the universal statement (ii)⇒(i) for all pairs over an infinite ring, the consistency across four structurally different choices – together with the structural argument already available from Theorem 2 (that $\mathrm{Ass}(M) = \{\mathfrak{p}\}$ forces $\sqrt{\mathrm{Ann}(N)} = \mathfrak{p}$ for every nonzero submodule $N \leq M$, from which (i) follows immediately) – corroborates the equivalence.

8 Verification of Theorem 6 (multiplication modules)

Theorem 6 (Theorem 6 of [1]). *Let R be a Noetherian commutative ring and M a finitely generated multiplication R-module with $\mathrm{Ass}(M) = \{\mathfrak{p}\}$. Then for all nonzero $N_1, N_2 \leq M$,*

$$\sqrt{\mathrm{Ann}(N_1 + N_2)} = \sqrt{\mathrm{Ann}(N_1)} + \sqrt{\mathrm{Ann}(N_2)} = \mathfrak{p}.$$

Recall that an R-module M is a multiplication module if every submodule of M has the form IM for some ideal $I \leq R$. Every cyclic module $M = R/I$ is trivially a multiplication module, since its submodules are exactly J/I for ideals $J \geq I$, i.e. $(J/I) \cdot (R/I)$. In particular, the module $M = R/(x^2)$ already used in Sections 4 and 7 is a multiplication module, and the computations recorded there – in which several pairs of cyclic submodules were shown to satisfy $\sqrt{\mathrm{Ann}(N_1 + N_2)} = \sqrt{\mathrm{Ann}(N_1)} + \sqrt{\mathrm{Ann}(N_2)} = (x)$ – already constitute a verification of Theorem 6 in the polynomial-ring setting, with the common value equal to the unique associated prime (x), exactly as the theorem asserts.

To complement this with the ring-theoretic setting of Example 3 of [1], we verify the identity directly and exhaustively on $M = \mathbb{Z}/25\mathbb{Z}$, whose complete submodule lattice is the chain $0 \subset 5\mathbb{Z}/25\mathbb{Z} \subset M$, hence small enough to check every pair of nonzero submodules explicitly.

```
n = 25
p = 5
submods = [5, 1]   # generators of 5Z/25Z and of M = Z/25Z itself

for a in submods, b in submods
    ann_a    = div(n, gcd(a, n))
    ann_b    = div(n, gcd(b, n))
    gen_sum  = gcd(a, b)
    ann_sum  = div(n, gcd(gen_sum, n))

    rad_is_p = (ann_sum % p == 0) && (ann_a % p == 0) && (ann_b % p == 0)
    println("Ann(N1)=", ann_a, "Z, Ann(N2)=", ann_b,
            "Z, Ann(N1+N2)=", ann_sum,
            "Z -- radical of all three equals p=", p, "Z? ", rad_is_p)
end
```

All four pairs $(a, b) \in \{5, 1\}^2$ (representing the two nonzero submodules $5\mathbb{Z}/25\mathbb{Z}$ and M itself, and their mutual combinations) return `true`: the annihilators involved are always $5\mathbb{Z}$ or $25\mathbb{Z}$, both of which have radical $5\mathbb{Z} = \mathfrak{p}$, so the common-value identity of Theorem 6 holds throughout the entire submodule lattice of M, confirming $\mathrm{Ass}(M) = \{5\mathbb{Z}\}$ from Example 3 of [1] directly at the level of every possible pair of submodules, rather than merely for the single generating example given there.

9 Conclusion

The computations above instantiate both directions of the equivalence in Theorem 1 of [1], the conclusion of Theorem 2, the contrapositive of Theorem 3, the hypothesis-and-conclusion pair of Theorem 4, the universal radical sum equivalence of Theorem 5, and the multiplication-module identity of Theorem 6, on genuinely non-principal polynomial, localized-ring, and cyclic integer-quotient data, independently of the hand computations given in the original proofs. Together with the direct reproduction of the elementary $\mathbb{Z}/n\mathbb{Z}$ examples, this provides a symbolic, machine-checked consistency check of all six of the paper's central characterizations. All source code is included above in full so that the computations can be reproduced by any reader with a working OSCAR installation.

References

[1] A. Pekin and H. Özkaya, *Annihilator-based dependency relations in modules and radical characterizations*, Algebra and Discrete Mathematics **41** (2026), no. 2, 235–243. https://doi.org/10.12958/adm2471

[2] The OSCAR Team, *OSCAR – Open Source Computer Algebra Research system*, Version 1.x, 2026. https://www.oscar-system.org